APL
PRESS
Box 27, Swarthmore, PA 19081

1 1

1976

APL NEWS

APL PRESS

Box 27, Swarthmore, PA 19081 (215) 328-9771

Dear friends,

APL Press is a new business formed for the purpose of publishing books, pamphlets, and other material related to APL.

This newsletter is the first of a series of occasional publications to be devoted to announcements and reviews of books, information about meetings, brief articles, problems and solutions, definitions of functions, correspondence, etc.

We would be pleased to receive manuscripts or outlines of projected books, material for the newsletter, comments and suggestions.

If you wish to continue to receive the newsletter please send us your name and address. We will also be glad to mail a copy of this issue to anyone you wish to suggest.

Sincerely,

Jean Iverson

APL WORKSHOP

An APL workshop was held May 16-18 at Queen's University, Kingston, Ontario, under the sponsorship of the Department of Computing and Information Science and the SIGPLAN Technical Committee on APL (STAPL). The 35 invited participants included representatives from some of the manufacturers who support APL (IBM, Burroughs, Digital Equipment Corporation), several time-sharing services and the academic community. The major topics discussed were the development of an APL standard, a standard information format for workspace interchange, and the need for language extensions in the areas of general arrays, control structures, and error handling and event control. A fuller report of the workshop activities and a list of addresses of the participants will be submitted to APL Quote Quad, or is available on request from the workshop organizer, M.A. Jenkins, Computing and Information Science, Queen's University, Kingston, Canada, K7L 3N6.

MAGIC CUBES

James G. Mauldon
Amherst College

The symbol `t` will be used to assert that the APL expression immediately following it is a well-defined logical array with no zero entry, so that `t`*EXPR* is equivalent to `t∧/,`*EXPR*. We shall use zero origin throughout.

We shall be concerned with arrays *C* satisfying the following three non-independent conditions:

```
(1)        t(ρC)=N,N,N        tC∊ιN*3        t(ιN*3)∊C
```

so that *C* is an arrangement of the elements of `ιN*3` in the form of a cube.

Our definition of a magic cube is far more stringent than that usually given. For us a magic cube of order *N* is an array *C* satisfying (1) and such that, in each of the thirteen possible directions, every set of *N* elements of *C* forming a straight line in the cube *C* has the same sum. In the usual definition, even when the cube is described as "pandiagonal", only seven directions are considered, namely the three orthogonal directions and the four directions of the space diagonals. Occasionally our magic cube has been referred to as a perfect magic cube.

A pandiagonal cube is an array *C* such that for any choice of integers *X*,*Y*,*Z* the array `Xφ[0]Yφ[1]Zφ[2]C` is a magic cube.

An associated cube is a magic cube *C* for which

```
(2)        t∧/(¯1+×/ρC)=(,C)+φ,C
```

The theory below provides a general method for the construction of magic cubes, including the following four examples:

```
(3)        (3ρ7)ρ7⊥7|(3 3↑3 5ρ4 5 6)+.×(3ρ7)⊤ι7*3
(4)    (3ρ8)ρ8⊥0 1 2 3 7 6 5 4[8|(3 3↑3 5ρ2 3 4)+.×(3ρ8)⊤ι8*3]
(5)   (3ρ9)ρ9⊥1 2 0 3 4 5 8 6 7[9|(3 3↑3 5ρ2 3 5)+.×(3ρ9)⊤ι9*3]
(6)       (3ρ11)ρ11⊥11|(3 3↑3 5ρ3 4 5)+.×(3ρ11)⊤ι11*3
```

Of these examples, (3) is associated, (4) is pandiagonal, and each of (5), (6) is an associated pandiagonal perfect magic cube.

THE THEORY

It is convenient to encode the elements of *C* and of ⍳*N**3 by using the matrices *MI* and *MC* defined below:

```
(7)        MI ← (3ρN)⊤⍳N*3          tMC=(3ρN)⊤,C
```

Observe that

```
(8)        t(⍳N*3)=N⊥MI          tC=(3ρN)ρN⊥MC
```

and that

```
(9)     tMI∊⍳N       tMC∊⍳N       t((ρMC)=ρMI)∧(ρMI)=3,N*3
```

Taking *K* (t3 3=ρ*K*) to be a suitably chosen matrix, subject to the constraints below, we define *C* by using (8) and (7) in conjunction with

```
(10)              MC ← N|K+.×MI
```

so that the array *C* is in fact defined by

```
(11)              C ← (3ρN)ρN⊥N|K+.×(3ρN)⊤⍳N*3
```

Then *C* will satisfy (1) if and only if *K* is invertible modulo *N*, which is equivalent to the condition

```
(12)              t1=N GCD DET K
```

where *DET* is the determinant and *GCD* is the greatest common divisor function.

The array *C* satisfying (1) will be a pandiagonal magic cube if *K* satisfies the condition

```
(13)           t∧/,1=N GCD K+.×1-(3ρ3)⊤⍳13
```

wherein columns 0 2 6 8 correspond to the four space diagonals.

The condition (2) for *C* to be an associated cube (if magic) will be satisfied if and only if

```
(14)                t∧/1=N|+/K
```

Since (13), (14) depend only and independently on the sets of numbers in each row of *K*, it is convenient to take *K* to be a circulant matrix:

```
(15)              K ← 0 2 1⌽3 3ρV
```

(which may also be written *K*←3 3↑3 5ρ*V*) where *V* (t3=ρ*V*) is a suitably chosen vector.

If t∧/0≠2 3 5]N and if we define

(16) K ← 3 3↑3 5ρ V ← (ι3)+ R ← (3]-N)×⌊N÷3

we find t1=N|+/V and t1=N|3×1+R so that K has the mod N inverse (3 3↑3 5ρ0 2 1×1+R) and consequently K satisfies (12) and (14) and C satisfies (1) and (2).

Since, except for columns 2 6 8, (13) is easily verified, it follows that in this case (t0≠2 3 5|N) (16) yields an associated magic cube, which is pandiagonal if also t0≠7|N. By taking N←7 or N←11 in (11) and (16) we obtain the examples (3) and (6).

If t∨/0=2 3 5 7|N it is impossible to satisfy (13), but we may obtain examples such as (4) and (5) by incorporating the step MC←W[MC] where W is a rearrangement of ιN such that, for any integers A and D,

(17) t0≠N|D implies t(+/W)=+/W[N]A+D×ιN]

and, in the case t0=3|N, by modifying V.

We close with seven problems, of which the last two are by far the most challenging.

CHALLENGES TO THE READER

A. Construct a magic cube of order 25.

B. Prove that t2044=+/,(X⌽U)/[0](Y⌽U)/[1](Z⌽U)/[2]C, where C is given by (4) and X,Y,Z,J are integers such that t2=+/U←(ι8)∊0,J. Interpret this result, and find other similar results.

C. Construct a four-dimensional analogue of a magic cube. Observe that the linear sums are the same in no fewer than forty different directions.

D. Construct a strongly associated magic cube of order 8, where a magic cube C is strongly associated if

(18) t∧/ ((ι×/ρC)=,C) ∨ ((ι×/ρC)=⌽,C)

E. Construct a 5×5×5 cube on ι125 in which more than 100 lines of length 5 have the same sum.

F. Determine whether or not there exists a magic cube of order 15

G. Determine whether or not there exists a four-dimensional magic cube whose order is less than 16.

αBET[?(?1E6)ρραBET]

What capital city was transformed to Eastern Capital by ¯2⌽ ?

APL programmer Bob Hankey of DSG, 133 South 36 Street, Philadelphia, now combines programming with the permanent repair of typeballs, a business he invented when his only APL typeball suffered a broken biscupid.

ON FUNCTION DEFINITION

A form of function definition particularly suited to exposition is defined and illustrated below in excerpts from Iverson's Elementary Analysis (APL Press, 1976). We would be pleased to publish any interesting and efficient APL program capable of fixing the definition of a function presented to it (as a character vector) in the αω form here described.

Consider a dyadic function *F* defined informally as follows: the result is the sum of the right argument and the product of the left argument and the square of the right argument. In other words, the result of the expression *X F Y* is *Y*+*X*×*Y*⋆2; specifically, 2 *F* 3 equals 3+2×3⋆2 or 21.

A formal definition is one which can be interpreted by a mechanical application of known rules, requiring no judgement or subtle interpretation. For example, the function *F* defined informally above can be defined formally as:

```
F:ω+α×ω⋆2
```

using the following rules: to interpret any expression of the form *X F Y*, substitute the first argument *X* for each occurrence of α (the first letter of the Greek alphabet) in the expression, and the last argument *Y* for each occurrence of ω (the last letter of the Greek alphabet). For example, the steps in the interpretation of the expression 4 *F* 3 can be shown as follows:

```
4 F 3
3+4×3⋆2
39
```

If the symbols α and ω do not both occur in a definition, the function defined is monadic. For example:

```
SQRT:ω⋆.5                    PITIMES:3.1416×ω
```

The colon in a function definition may be read aloud as "is". Thus, *F*:α+ω may be read as "*F* is α plus ω".

A variable which is assigned a value within a function definition is local to the function.

A function definition is said to be recursive if the function being defined recurs in the expression defining it. This notion may be familiar from informal definitions. For example, the power function *X*⋆*N* may be said to equal *X* times *X*⋆*N*-1, and the factorial function !*N* may be said to equal *N*×!*N*-1.

Let us attempt to define the factorial function *FAC* in this manner:

```
FAC:ω×FAC ω-1
```

To interpret the expression *FAC* 4 we would proceed by substitution as usual:

```
FAC 4
4×FAC 3
4×3×FAC 2
4×3×2×FAC 1
```

It is clear that this procedure can be terminated meaningfully only if we know the value of *FAC X* for some value of the argument *X*. In this case, *FAC* 1 is equal to 1, and with this knowledge we can terminate the interpretation as follows:

```
4×3×2×1
24
```

In general, it is necessary to know a second expression for the function (in this case the simple expression 1) and the condition under which it is to be applied (in this case when ω=1). The recursive definition of factorial therefore requires the following three pieces of information:

```
The primary expression:     ω×F ω-1
A proposition          :    ω=1
A secondary expression:     1
```

In a formal definition these three data are presented in the foregoing order with colons separating them. Thus:

```
      FAC:ω×FAC ω-1:ω=1:1

      FAC 4
24
```

Area of a closed figure. A figure is closed if the last point in its matrix representation equals the first. However, if a figure is known to be closed, it can also be represented more briefly by dropping the last point and requiring that the connection of the points be cyclic, i.e., the last connects to the first. Figure 5.19 shows the pentagon so represented by the following 5-column matrix:

```
      P
7 10  6  3  4
2  5  8  6  3
```

The area of a closed polygon so represented is given by the following single function:

```
      AREA:.5×+/(×⌿0 1⌽ω)-×⌿1 0⌽ω          [5.9.6]
```

Perimeter of a closed figure. If *P* is a two by *N* matrix representing the coordinates of the *N* vertices of a closed polygon, then the *N* displacements from vertex to vertex are given by the expression `D←P-¯1⌽P`. For example, if *P* is the matrix of Figure 5.19, then *N* is 5, and:

```
      P                  ¯1⌽P                  D
7 10  6  3  4        4  7 10  6  3        3  3 ¯4 ¯3  1
2  5  8  6  3        3  2  5  8  6       ¯1  3  3 ¯2 ¯3
```

Moreover, the length of any displacement (according to the theorem of Pythagoras) is the square root of the sum of the squares of the displacements along the *X* and *Y* axes. Consequently, the *N* lengths are given by `SQRT +⌿D*2`, where `SQRT:ω*.5`. The lengths of the sides of a polygon are therefore given by the following function applied to the matrix representation of its vertices:

```
LP:SQRT +⌿(ω-¯1⌽ω)*2                    [5.9.7]
```

The entire set of functions developed for handling polynomials is collected below:

```
P:+/α×ω*⍳Nα
N:+/ω=ω
PLUS:(ML↑α)+(ML←(Nα)⌈Nω)↑ω
MINUS:α PLUS-ω
TIMES:(α×1↑ω) PLUS 0,α TIMES 1↓ω:0=Nω:0
INTO:(1↑ω÷1↑α),α INTO 1↓ω MINUS α×1↑ω÷1↑α:(Nα)>Nω:⍳0
LINTO:⌽(⌽α) INTO ⌽ω
CURT:(-+/⌊\0=⌽ω)↓ω
PVRA:(α P 1↑ω),α PVRA 1↓ω:0=Nω:⍳0
```

Many of the properties of a function can be expressed in terms of the properties of the component functions used in its definition. Thus if `H:(Fω)+(Gω)`, then (as shown in Sec 2.3 and 3.3) the additive function for *H* is given by `AH:(αAFω)+(αAGω)`, and the derivative is `DH:(DFω)+(DGω)`. Rules for obtaining the derivatives of composite functions are listed below, using *DF* and *DG* to denote the derivatives of *F* and *G*, and *IF* and *IG* to denote their inverses:

NAME	COMPOSITE FORM	DERIVATIVE	
SUM	`(Fω)+Gω`	`(DF ω)+DG ω`	[6.1.1]
PRODUCT	`(Fω)×Gω`	`((Fω)×DG ω)+(DF ω)×Gω`	[6.1.2]
RECIPROCAL	`÷Fω`	`-(DF ω)÷(Fω)*2`	[6.1.3]
COMPOSITION	`F G ω`	`(DF Gω)×DG ω`	[6.1.4]
INVERSE	`IF ω`	`÷DF IF ω`	[6.1.5]
POWER	`ω*N`	`N×ω*N-1`	[6.1.6]

4.61 If *V* is any two-element vector, show that:

a)
```
2 ○ +/V
--/×/1 2∘.○ V
```

b)
```
1 ○ +/V
+/×⌿0 1⌽1 2∘.○ V
```